Tabelleninformationen per Abfrage und VBA ermitteln

Im Datenbankfenster können Sie in der Ansicht "Details" ablesen, wann eine Tabelle erstellt oder wann eine Tabelle das letzte Mal geändert wurde. Für verschiedene Anwendungen wie die Abstimmung von Datenbankversionen wäre es wichtig, diese Informationen auch per Abfrage bzw. VBA ermitteln zu können. Wenn dann noch zusätzlich die Anzahl vorhandener Datensätze ausgegeben werden könnte, lassen sich Abweichungen noch einfacher erkennen. Darüber hinaus benötigen Sie regelmäßig Informationen wie Feldnamen für die Dokumentation. Mit diesem zweiteiligen Tipp der Woche stellen wir Ihnen Abfragen und Funktionen vor, die Ihnen diese Informationen liefern.

Wenig bekannt ist die Möglichkeit, Informationen zu Tabellen aus der versteckten Systemtabelle "MSysObjects", in der Access Infos zu allen Datenbankobjekten festhält, per SQL-Anweisung auszulesen und Domainfunktionen auf diese Tabelle anzuwenden:

· Wechseln Sie im Datenbankfenster in den Bereich "Abfragen" und klicken Sie auf die Schaltfläche Neu. In Access 2007 klicken Sie im Register "Erstellen" in der Gruppe "Andere" auf Abfrageentwurf.

· Brechen Sie die Tabellenauswahl ab, indem Sie auf Schließen klicken.

· Stellen Sie die Abfrageansicht über das Menü Ansicht-Sql (bis Access 2003) bzw. per Klick auf Ansicht-SQL-Ansicht (Access 2007) auf den SQL-Modus um.

· Geben Sie im SQL-Eingabefenster die folgenden Anweisungen ein:

SELECT MSysObjects.Name, MSysObjects.DateCreate,
       MSysObjects.DateUpdate, 
       DCount("*",[Name]) AS Records
FROM MSysObjects
WHERE (((Left$([Name],4))<>"MSys") AND
      ((MSysObjects.Type)=1))
ORDER BY MSysObjects.Name;
· Speichern Sie die Abfrage unter dem Namen "Tabelleninfos" und verlassen Sie den Abfrageentwurf.

Wenn Sie die Abfrage nun im Datenbankfenster bzw. im Navigationsbereich per Doppelklick öffnen, erhalten Sie eine alphabetisch sortierte Übersicht der in der Datenbank vorhandenen Tabellen mit den Angaben zu Erstellungs- und Änderungsdatum sowie zur Anzahl der gespeicherten Datensätze.

Zur Funktionsweise der Abfrage: Sie selektiert zunächst alle Tabellen mit dem Kriterium 'Type= 1' und deren Erstellungs-/Änderungsdaten (DateCreate, DateUpdate) aus der Systemtabelle "MSysObjects", wobei Systemtabellen selbst durch das Kriterium 'Left$([Name],4)<>"MSys"' ausgeklammert werden. Gleichzeitig wird für jede Tabelle die Domainfunktion 'DCount("*",[Name]) AS Records' aufgerufen und so ein Feld "Records" mit der Anzahl Datensätze pro Tabelle generiert. Sortiert wird das Ganze schließlich nach dem Tabellennamen.

Mit dieser Abfrage können Sie sich schnell eine allgemeine Übersicht verschaffen. Um detaillierte Tabelleninfos auszulesen, in einer Tabelle zu speichern und diese Tabelle als Basis eines Berichtes für die Datenbankdokumentation zu verwenden, ist eine VBA-Funktion notwendig. Diese stellen wir in der nächsten Ausgabe von Access Weekly vor.

Im ersten Teil dieses Tipps hatten wir eine Abfrage vorgestellt, die eine alphabetisch sortierte Übersicht der Tabellen einer Datenbank mit den Angaben zu Erstellungs- und Änderungsdatum sowie zur Anzahl der enthaltenen Datensätze ausgibt. Um detaillierte Tabelleninfos auszulesen, in einer Tabelle zu speichern und diese Tabelle als Basis eines Berichtes für die Datenbankdokumentation zu verwenden, ist eine VBA-Prozedur notwendig, die wir im Folgenden vorstellen.

Zunächst ein paar Grundlagen: Informationen zu Tabellen speichert Access in einer Auflistung "TableDefs" des Objektes "Database". Jeder TableDef-Eintrag in dieser Auflistung verfügt neben der Eigenschaft "Name" mit dem Tabellenname über eine Auflistung "Fields" mit den Definitionen der einzelnen Felder. Hier sind die Eigenschaften "Name" (Name des Feldes), "Type" (Typ des Feldes) und "Size" (Feldgröße) von Interesse. Wenn im Tabellenentwurf unter "Beschreibung" etwas für ein Feld eingegeben wurde, kommt noch die Eigenschaft "Description" hinzu, die einer Auflistung "Properties" zu entnehmen ist.

Bevor wir zur Prozedur "CreateTableInfos" kommen, legen Sie zunächst eine Tabelle "TableInfos" mit den folgenden Feldern an:

Tabellenname: Datentyp "Text", Feldlänge 50
Feldname: Datentyp "Text", Feldlänge 50
Feldtyp: Datentyp "Text", Feldlänge 50
Feldgröße: Datentyp "Text", Feldlänge 50
Beschreibung: Datentyp "Text", Feldlänge 200

Danach kopieren Sie die folgende Prozedur in ein neues oder vorhandenes Modul:

Sub CreateTableInfos() 
  Dim X$
  Dim db As DAO.Database, rs As DAO.Recordset
  Dim tdf As DAO.TableDef, fld As DAO.Field

  On Error Resume Next
  DoCmd.SetWarnings False
  DoCmd.RunSQL "delete * from TableInfos;"
  DoCmd.SetWarnings True
  Set db = CurrentDb()
  Set rs = db.OpenRecordset("TableInfos", dbOpenDynaset)
  For Each tdf In db.TableDefs
    For Each fld In tdf.Fields
      rs.AddNew
      rs("Tabellenname") = tdf.Name
      rs("Feldname") = fld.Name
      X$ = "???"
      Select Case fld.Type
        Case 1: X$ = "Ja/Nein"
        Case 2: X$ = "Byte"
        Case 3: X$ = "Integer"
        Case 4
          If (fld.Attributes And 16) = 0 Then
            X$ = "Long Integer"
          Else
            X$ = "AutoWert"
          End If
        Case 5: X$ = "Währung"
        Case 6: X$ = "Single"
        Case 7: X$ = "Double"
        Case 8: X$ = "Datum/Zeit"
        Case 9: X$ = "Binary"
        Case 10: X$ = "Text"
        Case 11: X$ = "OLE-Objekt"
        Case 12
          If (fld.Attributes And 32768) = 0 Then
            X$ = "Memo"
          Else
            X$ = "Hyperlink"
          End If
        Case 15: X$ = "GUID"
        Case 16: X$ = "Big Integer"
        Case 17: X$ = "VarBinary"
        Case 18: X$ = "Char"
        Case 19: X$ = "Numeric"
        Case 20: X$ = "Decimal"
        Case 21: X$ = "Float"
        Case 22: X$ = "Time"
        Case 23: X$ = "Time Stamp"
        Case 101: X$ = "Attachment"
        Case 102: X$ = "Complex Byte"
        Case 103: X$ = "Complex Integer"
        Case 104: X$ = "Complex Long"
        Case 105: X$ = "Complex Single"
        Case 106: X$ = "Complex Double"
        Case 107: X$ = "Complex GUID"
        Case 108: X$ = "Complex Decimal"
        Case 109: X$ = "Complex Text"
      End Select
      rs("Feldtyp") = X$
      rs("Feldgröße") = fld.Size
      Err = 0
      X$ = fld.Properties("Description")
      If Err <> 0 Then X$ = "./."
      rs("Beschreibung") = X$
      rs.Update
    Next fld
  Next tdf

  rs.Close
  Set rs = Nothing
  Set db = Nothing

End Function

Um die Informationen zu Ihren Tabellen in die Tabelle "TableInfos" eintragen zu lassen, geben Sie im Direktbereich "CreateTableInfos" ein und drücken Return. Anschließend können Sie die Tabelle öffnen und durchsuchen oder als Basis eines Berichtes verwenden, den Sie der Datenbankdokumentation hinzufügen.

Die Prozedur "CreateTableInfos" löscht zunächst eventuell vorhandene Datensätze in der Tabelle "TableInfos" und öffnet sie dann als Recordset. Danach durchläuft sie in einer äußeren Schleife "For Each tdf..." die Tabellendefinitionen und in einer inneren Schleife "For Each fld..." die Felddefinitionen der jeweiligen Tabelle. Tabellenname, Feldname, Feldtyp und Feldgröße sowie ggf. eine vorhandene Beschreibung werden dabei in den entsprechenden Feldern des Recordsets gespeichert. Danach wird das Recordset wieder geschlossen und die Tabelle enthält die wichtigsten Informationen zu Ihren Tabellen.
